

CS 135

DESIGNING FUNCTIONAL PROGRAMS

DR. HOLBY • FALL 2014 • UNIVERSITY OF WATERLOO

Last Revision: December 6, 2014

Table of Contents

1	Introduction	1
1.1	Wise words from a wise teacher	1
1.2	Programming Language Design	1
1.3	Racket	1
1.4	Themes of the Course	2
2	Functional Programming	2
2.1	Introduction to Functional Programming	2
2.2	Function Definition	3
3	Software Design Recipe	4
3.1	Programs as Communication	4
3.2	Design Recipe Components	4
3.3	Suggested Procedure	4
3.4	Boolean-valued functions	5
3.5	Symbols	5
3.6	String manipulation	5
3.7	Conditions	6
3.8	Testing	6
4	Syntax and Semantics	6
4.1	Spelling rules	6
4.2	Substitution	7
5	Structure	8
5.1	Structure	8
5.2	Stepping	8
6	Lists	9
6.1	List Definition	9
6.2	Extracting from Lists	10
6.3	Recursion	11
6.4	Design Recipe	11
6.5	List Manipulation	11
6.6	Characters	11

7	Recursion	12
7.1	Processing Two Lists	13
7.2	Types of Recursion	14
8	Trees	14
8.1	Characteristics of Binary Trees	14
8.2	Binary Search Trees	15
9	Local Definitions	16
9.1	Semantics	16
9.2	Function Abstraction	17
9.3	Higher Order Functions	17
10	Graphs	18
10.1	Directed Acyclic Graphs	18
10.2	Cyclic Graphs	19
11	History	20

1 Introduction

1.1 Wise words from a wise teacher

- Keep up with the readings
- Take notes in lecture
- Start assignments early
- Get help early
- Follow our advice on approaches to writing programs (eg. design recipe, templates)
- Keep on top of workload
- Visit office hours to get help
- Integrate exam study into your weekly routine.
- Go beyond the minimum required
- Maintain a "big picture" perspective: Look beyond the immediate task or topic
- Go over your assignments and exams, learn from mistakes
- Read your mail sent to your UW email.

1.2 Programming Language Design

Imperative: based on frequent changes to data.

- Eg: machine language, Java, C++, Turing, VB

Functional: based on the computation of new values rather than the transformation of old ones.

- Examples: Excel formulas, LISP, ML, Haskell, Erlang, F#, Mathematica, XSLT, Clojure
- More closely connected to mathematics.
- Easier to design and reason about programs

1.3 Racket

- A functional programming language
- Minimal but powerful syntax
- Small toolbox with ability to construct additional required tools
- Interactive evaluator
- Used in education and research since 1975
- A dialect of Scheme

1.4 Themes of the Course

- Design (the art of creation)
- Abstraction (finding commonality, neglecting details)
- Refinement
- Syntax, expressiveness(understandability), and semantics (meaning)
- Communication

2 Functional Programming

2.1 Introduction to Functional Programming

Definition 2.1 (function). A **function** generalizes similar expressions and takes inputs to produce an output. These definitions consist of the **name** of the function, its **parameters**, and an **algebraic expression** using the parameters.

Definition 2.2 (application). An **application** supplies arguments for the parameters, which are substituted into the algebraic expression.

Example 2.1. If a function is defined as $g(x, y) = g(x + y)$, an application would be $g(1, 3)$

Example 2.2. We **evaluate** each of the arguments to yield values. One example of evaluating is substitution.

Definition 2.3 (substitution). Substitution must be done from innermost to outermost and then left to right. At each stage, only one substitution should be done. (If two evaluations have the same priority, the left most one is done first.)

Example 2.3. $f(x) = x^2$ $g(x, y) = x + y$

Incorrect: $g(g(1, 3), f(2)) = g(1, 3) + f(2)$ **Correct:** $g(g(1, 3), f(2)) = g(4, f(2))$

Example 2.4. If we treat infix operators $(+, -)$ like functions, parentheses are not required to specify order of operations. $3 - 2$ becomes $-(3, 2)$
 $(6 - 4)/(5 + 7)$ becomes $/(-(6, 4), +(5, 7))$

Example 2.5. In racket, the functions move inside the parentheses so $g(1, 3)$ becomes $(g 1 3)$
 $3 - 2 + 4/5$ becomes $(+(- 3 2)(/ 4 5))$

Note. Extra paratheses are harmful in Racket because they are used to signal a function application or other Racket syntax.

2.2 Function Definition

Functions

A function definition consists of a name for the function, a list of parameters, and a single "body" expression.

Example 2.6. Let $f(x) = x^2$ and let $g(x, y) = x + y$

In racket:

```
1 (define (f x) (* x x))
2 (define (g x y) (+ x y))
```

Each parameter name for the function has meaning only within the body if that function (local variable)

Example 2.7. For the following code:

```
1 (define (f x y) (+ x y))
2 (define (g x z) (* x z))
```

the two uses of x are independent and the two function definitions below define the same function.

```
1 (define (f x y) (+ x y))
2 (define (f a b) (+ a b))
```

Constants

Advantages:

- Give meaningful names to useful values
- Reduces typing and errors when such values are modified
- Enhances readability
- Can be used in any expression, including the body of function definitions.
- Sometimes referred to as variables, but value may not be changed.

Example 2.8. There are brackets around a function and its parameters to indicate that it is a function. To define a constant, no brackets are required around it.

Let $k = 3$ and $p = 3^2$

```
1 (define k 3)
2 (define p (* k k))
```

3 Software Design Recipe

3.1 Programs as Communication

Communication occurs between:

- You and the computer
- You and yourself in the future (;; Comments)
- You and others

For software design, programs should be: compatible, composable, correct, durable, efficient, extensible, flexible, maintainable, portable, readable, reliable, reusable, scalable, usable, and useful.

3.2 Design Recipe Components

1. **Purpose:** Describes what the function is to compute.
2. **Contract:** Describes what type of arguments the function consumes and what type of value it produces. (Num, Int, Nat (includes 0), Any, Boolean)
3. **Examples:** Illustrating the typical use of the function.
4. **Definition:** The Racket definition (header and body) of the function.
5. **Tests:** A representative set of inputs and expected outputs.

3.3 Suggested Procedure

Draft → Examples → Definition Header & Contract → Parameters → Definition → Tests.

Example 3.1. An example of a well formatted design recipe:

```
1 ;; (sum-of-squares p1 p2) produces the sum of
2 ;; the squares of p1 and p2
3 ;; sum-of-squares: num num -> num
4 ;; examples
5 (check-expect (sum-of-squares 3 4) 25)
6 (check-expect (sum-of-squares 0 2.5) 6.25)
7
8 (define (sum-of-squares p1 p2)
9   (+ (* p1 p1) (* p2 p2)))
10
11 ;; Tests
12 (check-expect (sum-of-squares 0 0) 0)
13 (check-expect (sum-of-squares -2 7) 53)
```

If there are additional constraints on the inputs, the contract section may be extended to add a requires section.

```

1 ;; (my-function a b c)
2 ;; my-function: num num num -> num
3 ;; requires: 0 < a < b
4 ;;           c > 0

```

3.4 Boolean-valued functions

Definition 3.1. A **predicate** is a function that produces a boolean result. They are written in the form

```

1 ;;(operator a b ...)
2 ;;examples
3 (> a b)
4 (= a b)

```

Standard racket outputs `#t` and `#f` and these values can be used in `cond` statements and other functions that take in boolean values.

Example 3.2. You can use `equal?` to test the equality of two values which may not be of the same type.

Note. A predicate can also be user-defined and can use compound operators such as AND, OR, NOT.

Example 3.3. An example of a predicate that outputs if a number is between two numbers

```

1 (define (between? low high numb)
2   (and (< low numb) (< numb high)))

```

3.5 Symbols

Definition 3.2. A **symbol** can be used to represent something. It is defined using a single apostrophe, 'blue. To compare symbols, use `symbol=?`.

```

1 (define something 'blue)
2 (symbol=? something 'blue)

```

3.6 String manipulation

There are many string manipulation techniques in the manual that are not covered in the lectures. Consult the manual for more information.

```

1 (string? "Warbear") ;determines whether argument is string (#t)
2 (string-append a b) ;merges strings (ab)
3 (string-length x) ; yield the length of string x(1)
4 (string<? "Apple" "apple") -> #t
5 ;;determines whether first string is lexicographically less than the
   second.

```

```
6 (string<? "apple" "Apple") -> #f ;;lower case comes after upper case, so
   a > A
7 (string<? "a" "b" "c") -> #t
```

Definition 3.3. Substring returns a new string and must be less than or equal to the length of the original string. The end value must be greater than or equal to start. Returns the substring from the start character to the end character, but not including the end character. The end character is an optional parameter. If none specified, assume end is last character.

Note. Strings start at 0 in computer science.

```
1 (substring "Semantics" 2 5) -> "man"
2 (substring "Semantics" 2) -> "mantics"
3 (substring "Syntax" 3 3) -> "" (empty string)
```

3.7 Conditions

Similar to if statements in other languages, and are eventuated from top to bottom. They can also be nested provided that the required parameters are filled.

```
1 (cond
2   [boolean action]
3   [boolean action]
4   [else action])
```

3.8 Testing

Definition 3.4. Black-box testing is testing done when the content of the code is unknown or not required.

Definition 3.5. White-box testing is testing done when content of code is known and required.

4 Syntax and Semantics

4.1 Spelling rules

Identifier names are made up of letters, numbers, hyphens, and underscores. Must contain at least one non-number and no brackets or apostrophes. Symbols start with a single quote followed by a valid identifier name.

Definition 4.1. Syntax is the way that code is allowed to be written. Usually code will not compile without proper syntax.

Definition 4.2. Semantics is the meaning of what is said.

Definition 4.3. A valid programming statement must have exactly one meaning to avoid **ambiguity**

4.2 Substitution

Racket evaluates functions through substitution. Only one substitution can be made at a time.

Definition 4.4. A step-by-step reduction in accordance to the rules is called **tracing** a program.

Example 4.1. **cond** statements skip false statements and goes to the first true statement.

```
1 (cond [false exp] ...) -> (cond ...)  
2 (cond [true exp]... -> exp  
3 (cond [else exp]... -> exp
```

Example 4.2. **and** statements evaluate to false if one argument is false. True skips to the next argument, and a stand alone (and) must be reached for the statement to be true.

```
1 (and false...) -> false  
2 (and true...) -> (and ...)  
3 (and) -> true
```

Example 4.3. **or** statements are the opposite of and statements. Evaluates to true if one argument is true, and false skips to the next argument. A stand alone (or) is substituted with false.

```
1 (or true ...) -> true  
2 (or false ...) -> (or...)  
3 (or) -> false.
```

Constants are directly replaced by their actual value even if other constants are called. It is all done in one substitution step.

Definition 4.5. A **syntax error** occurs when a sentence cannot be interpreted using the programming language's grammar and convention.

Definition 4.6. A **run-time error** occurs when an expression cannot be reduced to a value, but the evaluation steps are still incomplete.

```
1 (cond [(> 3 4) x] ;no else statement, cond is false
```

5 Structure

5.1 Structure

Definition 5.1. Structures bundle data together, and are similar to objects and classes in other languages.

Posn

Posn is a predefined structure. Various functions are included when a structure is defined.

```
1 ;;A Posn is a (make-posn x y)
2 (make-posn 1 1) ;;makes a posn object with the given arguments
3 (posn-x posnname) ;; returns x value of a posn
4 (posn-y posnname) ;; returns y value of a posn (can be extended when
   structure has more arguments)
```

Note. Since posn is a data structure, (make-posn 8 1) cannot be reduced any further and is considered a value.

Definition 5.2. The **scale** function scales the posn by a given factor.

```
1 (check-expect (scale (make-posn 6 9 2)) (make-posn 12 18))
```

Definition 5.3. Racket uses **dynamic typing** and one cannot enforce a parameter to be a specific value type.

```
1 (make-posn 'shadow 'hawker) ;valid, racket does not enforce contracts
```

Structure Definition

Structures can be defined using define-struct. Various functions are also created when the structure is defined.

```
1 (define-struct card (suit num)) ;A Card is a (make-card Sym Nat)
2 (make-card 'Hearts 7)           ;makes a card
3 (card-num (make-card 'Hearts 7)) ;returns num parameter of card (7)
4 (card? (make-card 'Spades 10))  ;checks whether argument is a card
```

5.2 Stepping

Using code from the previous example,

```
1 (card-num (make-card 'Hearts 7)) -> 7
2 (card? 5) -> false
```

Definition 5.4. A **template** is a function that consumes a structure and outputs every argument within the structure.

```

1 ;; my-mp3info-fn: Mp3Info -> Any
2 (define (my-mp3info-fn info)
3   (... (mp3info-performer info)...
4     (mp3info-title info)...
5     (mp3info-length info)...
6     (mp3info-genre info)...))

```

Definition 5.5. **anyof** is helpful in the contract section if a function can return different data structure outputs based on the input.

```

1 ;;divide: Num Num -> (anyof Num Boolean)
2 (define (divide x y)
3   (cond [(= y 0) false]
4     [else (/ x y)]))

```

6 Lists

6.1 List Definition

Definition 6.1. A **list** is a recursive structure defined in terms of a smaller list. A list includes one item in the list, and a sublist not including that item. It is very similar to a linked list in other CS languages. The data definition is:

```

1 ;; A List is one of:
2 ;; * empty
3 ;; * (cons Any List)

```

The template is:

```

1 ;; my-list-fn: (listof Any) -> Any
2 (define (my-list-fn list)
3   (cond
4     [(empty? list) ...]
5     [else ... (first list) ...
6       ... (my-list-fn (rest list))]))

```

Definition 6.2. The function **cons** is used to take an element and a list and produces a new, longer list.

```

1 (cons a b)

```

where a represents an item in the list and must be a value (Str, Num, Sym, etc), and b must be a list (empty is a list). Similar to posn, (cons a b) is in simplest form.

Example 6.1. A few examples of defining lists.

```

1 (define clst empty) ;;empty list ; Empty list
2 (define clst1 (cons 'Heavy empty)) ; A list containing one item.
3 (define clst2 (cons 'Light clst1)) ;A list containing two items, Light
  and Heavy
4 (define clst3 (cons 'Light (cons 'Heavy empty))) ;c2 and c3 are the same

```

6.2 Extracting from Lists

Definition 6.3. The **first** function returns the first element in a given list.

```

1 (first (cons a b)) --> a ;substitution process
2 ;;Returns first element in the list (a).

```

Definition 6.4. The **rest** function returns the list of elements without the first.

```

1 (rest (cons a b)) -> b ;substitution process
2 ;;Returns the list b

```

Note. With a very long list of items, first and rest can be used together to produce a chosen element.

```

1 (define somelist (cons a (cons b (cons c (cons d (cons e empty))))))
2 (first (rest (rest somelist))) ;produces c

```

Definition 6.5. The **empty** function returns true and false depending on whether a given list is empty. For substitution, #t or #f is substituted in one step.

```

1 (empty? (cons 'Food empty)) ; #t if list empty, #f if not. (#f)

```

Definition 6.6. The **member** function returns true and false depending on whether a given element is in a given list.

```

1 (member? 'Food (cons 'Food empty)) ; (#t)

```

Definition 6.7. The **length** function returns the length of a given list.

```

1 (length los) ;returns length of los
2
3 ;;Without the built-in function, this would be the method to count the
  length.
4 (define (count-concerts los)
5   (cond [(empty? los) 0]
6     [else (+ 1 (count-concerts (rest los)))]))

```

6.3 Recursion

Definition 6.8. A function is **recursive** if the body of a function requires the application of the same function.

Definition 6.9. In **structural recursion** all parameters to the recursive call are either unchanged or one step closer to a base case.

6.4 Design Recipe

The design recipe for lists will contain a self-referential data definition.

```

1 ;; A List-Of-Symbols is one of:
2 ;; * empty
3 ;; * (cons Sym List-Of-Symbols)
4
5 ;; my-los-fn: (listof Sym) -> Any
6 (define (my-los-fn los)
7   (cond [(empty? los)... ]
8         [else (... (first los)...(my-los-fn (rest los))...)]))

```

6.5 List Manipulation

Example 6.2. A function that negates every number in a list, and produces a new list.

```

1 ;; (negate-list lon) negates every number in lon
2 ;; negate-list: (listof Num) -> (listof Num)
3 ;; examples:
4 (check-expect (negate-list empty) empty)
5 (check-expect (negate-list (cons 2 (cons -12 empty)))
6 (cons -2 (cons 12 empty)))
7 (define (negate-list lon)
8   (cond [(empty? lon) empty]
9         [else (cons (- (first lon)) (negate-list (rest lon)))]))

```

Note. Sometimes, functions only make sense if it is given a nonempty list. An additional statement can be added to the requires section of the design recipe to indicate this.

6.6 Characters

Definition 6.10. A **string** is just a sequence of characters. Racket's notation for a character 'a' is #\a

Definition 6.11. The function **string->list** converts a given string into a list of characters.

```

1 (string->list "test")
2 ;;The code below is the output of the above function.
3 (cons #\t (cons #\e (cons #\s (cons #\t empty))))

```

Example 6.3. An example of where this would be useful is counting characters in a string.

```

1 (define (vowel? c)
2   (member c(string->list "aeiouAEIOU")))
3
4 (define (count-vowels-list loc)
5   (cond [(empty? loc) 0]
6         [(vowel? (first loc)) (+ 1 (count-vowels-list (rest loc)))]
7         [else (count-vowels-list (rest loc))]))
8
9 (define (count-vowels str)
10  (count-vowels-list (string->list str)))

```

Note. Lists and structures can be used together to create lists of structures which can make things pretty complicated.

7 Recursion

Example 7.1. An example of a countdown function using recursion.

```

1 (define (countdown n)
2   (cond [(= n 0) ...]
3         [else (cons n (countdown (sub1 n)))]))

```

If it is needed for the countdown to stop at a desired number, another parameter can be added. The parameter will "go along for the ride" and be called in every iteration.

Example 7.2. Using recursion, a list can be easily sorted. **insert** is a recursive function which takes in a number and a sorted list, and inserts the number into the list.

```

1 (define (sort lon)
2   (cond [(empty? lon) empty]
3         [else (insert (first lon) (sort(rest lon)))]))
4
5 (define (insert n slon)
6   (cond [(empty? slon) (cons n empty)]
7         [(<= n (first slon)) (cons n slon)]
8         [else (cons (first slon) (insert n (rest slon)))]))

```

Definition 7.1. Lists can be abbreviated. However, lists have fixed sizes while cons can contain a list of any arbitrary size.

```

1 (cons exp1 (cons exp2 empty))
2 ;; becomes
3 (list exp1 exp2).

```

Definition 7.2. The commands **(second my-list)** is equivalent to **(first (rest my-list))**. This pattern goes all the way up to **(eighth (my-list))**. Use these sparingly to improve readability.

Definition 7.3. Lists can be further abbreviated by **quoting**. For instance `(cons 'red (cons 'blue empty))` can be written as `'(red blue)` with the apostrophe being outside the list. Similarly, numbers can use the quoted list format too because quoted numbers evaluate to numbers.

Definition 7.4 (AL). An **association list** is a list that contains a list of (key, value) pairs. An example would be a **dictionary**.

```
1 ;; my-al-fn: AL -> Any
2 (define (my-al-fn alst)
3   (cond
4     [(empty? alst) ...]
5     [else ... (first (first alst))...
6       ... (second (first alst))...
7       ... (my-al-fn (rest alst))]))
```

Example 7.3. Example of AL:

```
1 ;; (lookup-al k alst) produces the value corresponding to key k,
2 ;; or false if k not present
3 ;; lookup-al: Num AL -> (anyof Str false)
4 (define (lookup-al k alst)
5   (cond [(empty? alst) false]
6         [(equal? k (first (first alst))) (second (first alst))]
7         [else (lookup-al k (rest alst))]))
```

7.1 Processing Two Lists

Example 7.4. Processing **one list** is simple because the other list will just "go along for a ride".

Example 7.5. Processing in **lockstep** requires that the lists are the same length and consumed at the same rate.

```
1 ;; dot-product: (listof Num) (listof Num) -> Num
2 (define (dot-product lon1 lon2)
3   (cond [(empty? lon1) 0]
4         [else (+ (* (first lon1) (first lon2))
5                   (dot-product (rest lon1) (rest lon2)))]))
```

Example 7.6. Processing at **different rates** is more complicated and requires at least four conditions.

```
1 (define (merge lon1 lon2)
2   (cond [(and (empty? lon1) (empty? lon2)) empty]
3         [(and (empty? lon1) (cons? lon2)) lon2]
4         [(and (cons? lon1) (empty? lon2)) lon1]
5         [(and (cons? lon1) (cons? lon2))
6          (cond [(< (first lon1) (first lon2))
7                  (cons (first lon1) (merge (rest lon1) lon2))]
8                    [else (cons (first lon2) (merge lon1 (rest lon2)))]))]))
```

Note. In some cases, conditions can be combined if they produce the same output.

7.2 Types of Recursion

Definition 7.5. Structural recursion, all parameters to the recursive call are either unchanged or one step closer to a base case.

Definition 7.6. Structural recursion with an accumulator is when a parameter called the **accumulator** is passed down. An example would be finding the largest element in a list.


Definition 7.7. When the arguments in the recursive application are **generated** and the function does not follow an inductive approach (breaking out of a recursion by reaching a base case), then a function is following **generative recursion**. Parameters are freely calculated at each step.

```
1 (define (euclid-gcd n m)
2   (cond [(zero? m) n]
3         [else (euclid-gcd m (remainder n m))]))
```

8 Trees

Definition 8.1. A **binary arithmetic expression** consists of numbers joined by binary operators.

Example 8.1.



8.1 Characteristics of Binary Trees

- Internal nodes have exactly two children
- Leaves have number labels while internal nodes have symbol labels
- Order of the children matter
- Structure of tree dependent on given expression

Example 8.2. Structure for Binary Trees (3rd line should be in comments):


```

1 (define-struct binode (op arg1 arg2))
2 ;; A Binary arithmetic expression Internal Node (BNode)
3 ;; is a (make-binode (anyof (+ - * /) BinExp BinExp)
4
5 ;; A binary arithmetic expression (BinExp) is one of:
6 ;; * a Num
7 ;; * a BNode

```

Definition 8.2. Evaluation of binary trees requires the recognition of all the symbols

```

1 (define (eval ex)
2   (cond [(number? ex) ex]
3         [else (eval-binode (binode-op ex)
4                             (eval (binode-arg1 ex))
5                             (eval (binode-arg2 ex)))]))
6
7 (define (eval-binode op left right)
8   (cond [(symbol=? op '-') (- left right)]
9         [(symbol=? op '+) (+ left right)]
10        [(symbol=? op '/') (/ left right)]
11        [(symbol=? op '*') (* left right)]
12        [else (error "Unrecognized operator")]))

```

8.2 Binary Search Trees

Definition 8.3. Binary search trees are structures that store data and makes searching for items more efficient. This requires that for each node, all the elements in its left subtree are less than it, and elements in the right subtree are greater than it. If a node has no children, its left and right subtrees are denoted as empty.

```

1 (define-struct node (key val left right))
2 ;; A Node is a (make-node Num Str BST BST)
3
4 ;; A Binary Search Tree (BST) is one of:
5 ;; * empty
6 ;; * a Node

```

Aside. In a traditional list, to find an item, you go through each element one by one and checking equality, leading to a worst case scenario of $O(n)$. In a well-balanced binary search tree, every traversal cuts off half of the elements, leading to a time complexity of $O(\log n)$.

Example 8.3. Checking if an item is in a tree:

```

1 (define (in-tree x tree)
2   (cond
3     [(empty? tree) false] ;If tree is empty, false
4     [(equal? x (node-value tree)) true] ;Element is at the node
5     [else (or (in-tree x (node-left tree)) (in-tree x (node-right
      tree))))]) ;; Checks both subtrees

```

Note. The above template can be modified if certain computations are required with the value.

9 Local Definitions

Definition 9.1. Local constants and variables may only be used within a local block. This is useful for larger programs where a variable name is not needed outside a certain area.

Note. When a variable name is both defined in global and local, the local version takes priority within the lexical scope. In CS 135, local functions do not require examples or tests.

```

1 ;; Heron Formula
2 (define (t-area4 a b c)
3   (local [(define s (/ (+ a b c) 2))]
4     (sqrt (* s (- s a) (- s b) (- s c)))))

```

9.1 Semantics

When a local variable or function is defined, it is first promoted to the top level, and the new name is substituted everywhere the old name is used in the expression. In the next example, a random variable name **s₄₇** is generated.

```

1 (t-area4 3 4 5)
2 ;;Function is called
3 (local [(define s (/ (+ 3 4 5) 2))]
4   (sqrt (* s (- s 3) (- s 4) (- s 5))))
5 ;;Local is moved outside
6 (define s_47 (/ (+ 3 4 5) 2))
7 (sqrt (* s_47 (- s_47 3) (- s_47 4) (- s_47 5)))
8 ;;New variable is first simplified
9 (define s_47 6)
10 (sqrt (* s_47 (- s_47 3) (- s_47 4) (- s_47 5)))
11 ;;Variable is substituted in
12 (sqrt (* 6 (- 6 3) (- 6 4) (- 6 5)))

```

Example 9.1. Sometimes local can be used for clarity. The distance function may appear longer, but it becomes easier to understand.

```

1 (define (distance posn1 posn2)
2   (sqrt (+ (sqr (- (posn-x posn1) (posn-x posn2)))
3            (sqr (- (posn-y posn1) (posn-y posn2))))))

```

```

1 (define (distance posn1 posn2)
2   (local [(define delta-x (- (posn-x posn1) (posn-x posn2)))
3           (define delta-y (- (posn-y posn1) (posn-y posn2)))]
4     (sqrt (+ (sqr delta-x) (sqr delta-y)))))

```

9.2 Function Abstraction

Definition 9.2. Abstraction is the process of finding similarities and forgetting unimportant differences. Advantages of functional abstraction include:

- Reduced code size
- Avoided cut-and-paste
- Bugs may be fixed on one place

Example 9.2. $(+ \ x \ y)$ and $(* \ x \ y)$ are similar if you ignore the function. Furthermore, they can be abstracted into one function that consumes a function and two numbers. X s are used in the contract to signal that the parameters are of the same type.

```

1 ;;foo: (X X -> X) Num Num -> Num
2 (define (foo f x y) (f x y))
3
4 (foo + 2 3) ; 5
5 (foo * 2 3) ; 6

```

Example 9.3. Similarly a function can produce a function that could be applied to other arguments.

```

1 (define add3 (local [(define (f m) (+ 3 m))] f))
2
3 (add3 2) ;;Produces 5

```

Definition 9.3. Lambda may be used to create a function. This is a much cleaner than using a local block.

```

1 (define add3 (lambda (x) (+ 3 x)))

```

9.3 Higher Order Functions

Definition 9.4. The Racket built-in function **filter** consumes a predicate with contract $X \rightarrow \text{Bool}$ and a list, and produces a list of all the elements in the list that satisfy that condition.

```

1 ;; filter: (X -> Bool) (listof X) -> (listof X)
2
3 (filter (lambda (x) (= (modulo x 2) 0)) '(1 2 3 4))
4 (filter even? '(1 2 3 4))
5 ;; Both filters produce (list 2 4)

```

Definition 9.5. The Racket built-in function **map** consumes a function and a list, and applies the function to each element in the list, outputted in a new list.

```

1 ;; map: (X -> Y) (listof X) -> (listof Y)
2
3 (map add1 '(6 9)) ;; produces (list 7 10)

```

Note. **map** may consume an arbitrary amount of lists, as long as all the parameters are dealt with either by lambda or by another function.

Definition 9.6. The Racket built-in function **foldr** consumes a two argument function, and applies the operation to the base case, last element, second last element, and so on until the first item of the list.

```

1 ;;foldr
2 (foldr (lambda (x y) (+ x y)) 7 (1 2 3)) ;;produces 7 + 3 + 2 + 1 = 13

```

Example 9.4. All duplicates may be replaced from a list using foldr and no explicit recursion.

```

1 (define (dedup list)
2   (foldr (lambda (element newlist)
3           (cond [(member? element newlist) newlist] ; Already exists
4                 [else (cons element newlist)])) ; Unique element
5   empty list))

```

10 Graphs

10.1 Directed Acyclic Graphs

Definition 10.1. A **directed graph** consists of a collection of **nodes** together with a collection of **edges**.

Definition 10.2. An **edge** is an ordered pair of nodes, represented by an arrow from the first node to the second. Given an edge (v, w) , w is an **out-neighbour** of v , and v is an **in-neighbour** of w .

Definition 10.3. A graph has a **cycle** if following all the edges results in an infinite loop.

Definition 10.4. Directed graphs without cycles are called **directed acyclic graphs**.

```

1 ;;A Node is a Sym
2 ;; A Graph is a (listof (list Node (listof Node)))

```

Example 10.1. A useful operation for a graph includes finding its neighbours.

```

1 ;; (neighbours v G) produces list of neighbours of v in G
2 ;; neighbours: Node Graph -> (listof Node)
3 (define (neighbours v G)
4   (cond [(empty? G) (error "vertex not in graph")]
5         [(symbol=? v (first (first G))) (second (first G))]
6         [else (neighbours v (rest G))]))

```

Example 10.2. To find a route between two points in a graph, mutual recursion may be used to generate a list of neighbours, and keep finding neighbours until either the point is found or no neighbours are left.

```

1 ;; (find-route/list los dest G) produces route from
2 ;; an element of los to dest in G, if one exists
3 ;; find-route/list: (listof Node) Node Graph -> (anyof (listof Node)
4 ;; false)
5 (define (find-route/list los dest G)
6   (cond [(empty? los) false]
7         [else (local [(define route (find-route (first los) dest G))]
8                 (cond [(false? route)
9                       (find-route/list (rest los) dest G)]
10                  [else route]))]))

```

```

1 ;; (find-route orig dest G) finds route from orig to dest in G if it
2 ;; exists
3 ;; find-route: Node Node Graph -> (anyof (listof Node) false)
4 (define (find-route orig dest G)
5   (cond
6     [(symbol=? orig dest) (list orig)]
7     [else (local [(define nbrs (neighbours orig G))
8                   (define route (find-route/list nbrs dest G))]
9             (cond [(false? route) route]
10                  [else (cons orig route)])))]))

```

10.2 Cyclic Graphs

Note. For **cyclic graphs**, this function will never terminate. To guarantee termination, create another parameter called **visited**, and ensure that each node has not been already visited.

```

1 ;; find-route/list: (listof Node) Node Graph (listof Node)
2 ;; -> (anyof (listof Node) false)
3 (define (find-route/list los dest G visited)
4   (cond [(empty? los) false]
5         [(member? (first los) visited)
6          (find-route/list (rest los) dest G visited)]
7         [else (local [(define route (find-route/acc (first los)
8                                                       dest G visited))]
9                   (cond [(false? route)
10                          (find-route/list (rest los) dest G visited)]
11                          [else route]))]))

```

The code for find-route/list does not change the accumulator, it only uses it. Adding the accumulator is done in the next code snippet.

```

1 ;; find-route/acc: Node Node Graph (listof Node)
2 ;; -> (anyof (listof Node) false)
3 (define (find-route/acc orig dest G visited)
4   (cond [(symbol=? orig dest) (list orig)]
5         [else (local [(define nbrs (neighbours orig G))
6                          (define route (find-route/list nbrs dest G
7                                                         (cons orig visited)))]
8                 (cond [(false? route) route]
9                       [else (cons orig route)]))]))

```

11 History

Example 11.1. Charles Babbage did meechanical computattion for military applications, but the computational operations were separate from the execution.

Example 11.2. Ada Augusta Byron assissted Babbage in explaining and promoting his ideas. She was possibly the first computer scientist.

Example 11.3. David Hilbert formalized the asiomatic treatment of Euclidean geometry. He believed that these three questions can be answered:

- Is mathematics complete? (For any formula ϕ , if ϕ is true, ϕ is provable.)
- Is mathematics consistent? (For any formula ϕ , thehre aren't proofs for both ϕ and $\neg\phi$.)
- Is there a procedure to, given a formula ϕ , produce a proof of ϕ , or show there isn't one?

Example 11.4. Kurt Godel proved that any non-trivla system of axioms (a system of axioms capable of describing integer arithmetic) is not complete. If it is consistent, it cannot be proved with the system. "The statement cannot be proved" would be inconsistent if the statement was false, because it would be a proof of a false statement, so it must be true but not provable.

Example 11.5. Alonzo Church and his student Kleene invented lambda calculus.

$\lambda x, x + 2$ is a function that adds 2 to its argument

Example 11.6. Alan Turing defined a different model of computation and resulted in a simpler and influential proof.

Example 11.7. Grace Murray Hopper was the author of the first compiler defining an english-like data processing language, which later became COBOL.

Example 11.8. John Backus designed FORTRAN, which became the dominant language of numerical and scientific computation. He proposed a functional programming language, which led to the development of Lisp.

Example 11.9. John McCarthy was an AI researcher at MIT known for designing and implementing Lisp, based on ideas from lambda calculus and recursive function theory.

LISP eventually evolved into a general purpose programming language, and became the dominant language in AI due to encouragement of modifying the language itself. It came two main standards after the 1980s: Common Lisp and Scheme.

Example 11.10. Gerald Sussman invented Scheme after an idea for a Lisp-like programming language had actors and lexical scoping added. He also wrote the textbook “Structures and Interpretation of Computer Programs”, based on Scheme.

Example 11.11. The textbook “How to Design Programs” was written to remedy some issues with that book, namely the steep learning curve and lack of methodology.